

**BEFORE BREAK PROCESSING**

BEFORE [BREAK] [PROCESSING] { statement(s) DO statement(s) DOEND }
END-BEFORE

- Used to initialize user-defined fields used in AT BREAK processing.
- Used in conjunction with automatic break processing to perform processing after the record has been read but *before*:
  - Control is checked for AT BREAK condition;
  - Statements specified with AT BREAK are executed;
  - Any Standard System Functions specified are evaluated
- Allowed once in FIND, READ, HISTOGRAM, SORT or READ WORK FILE loops.
- Must have an AT BREAK which applies to the loop.
- END-BEFORE required in Structured Mode.
- Must be placed somewhere within the processing loop.
- »» Best used in conjunction with automatic break processing when an ACCEPT or REJECT statement is also present in a program (see the example below).

```

0010 ** BEFORE BREAK PROCESSING EXAMPLE
0020 DEFINE DATA LOCAL
0030 1 JOBS VIEW OF EMPLOYEES
0040 2 NAME
0050 2 JOB-TITLE
0060 2 CURR-CODE(1)
0070 2 SALARY(1)
0080 2 BONUS(1,1)
0090 1 #TOT-SALARY(P11)
0100 END-DEFINE
0110 READ JOBS BY JOB-TITLE STARTING FROM 'A' THRU 'AZ'
0120  ACCEPT IF SALARY(1) GT 0
0130  DISPLAY NOTITLE (SG=OFF) NAME SALARY(1)
0140  BONUS(1,1) 'TOTAL/SALARY' #TOT-SALARY
0150  COMPUTE #TOT-SALARY = SALARY(1) + BONUS(1,1)
0160  AT BREAK OF JOB-TITLE
0170  PRINT 'POSITIONS OF' JOB-TITLE
0180  COUNT(JOB-TITLE) SUM(#TOT-SALARY) /
0190  END-BREAK
0200 END-READ
0210 END

```

- In this example, a record that potentially meets the ACCEPT logic could trigger a break situation which would be incorrect. The BEFORE BREAK PROCESSING statement provides a means of utilizing the ACCEPT without corrupting the AT BREAK function.

**BEFORE BREAK PROCESSING**

```

0010 ** BEFORE BREAK PROCESSING EXAMPLE
0020 DEFINE DATA LOCAL
0030 1 JOBS VIEW OF EMPLOYEES
0040 2 NAME
0050 2 JOB-TITLE
0060 2 CURR-CODE(1)
0070 2 SALARY(1)
0080 2 BONUS(1,1)
0090 1 #TOT-SALARY(P11)
0100 END-DEFINE
0110 READ JOBS BY JOB-TITLE STARTING FROM 'A' THRU 'AZ'
0120 ACCEPT IF SALARY(1) GT 0
0130 DISPLAY NOTITLE (SG=OFF) NAME SALARY(1)
0140 BONUS(1,1) 'TOTAL SALARY' #TOT-SALARY
0150 COMPUTE #TOT-SALARY = SALARY(1) + BONUS(1,1)
0160 BEFORE BREAK PROCESSING
0170 ACCEPT IF SALARY(1) GT 0
0180 END-BEFORE
0180 AT BREAK OF JOB-TITLE
0190 PRINT 'POSITIONS OF' JOB-TITLE
0200 COUNT(JOB-TITLE) SUM(#TOT-SALARY) /
0210 END-BREAK
0220 END-READ
0230 END
    
```

MORE _	NAME	ANNUAL SALARY	BONUS	TOTAL SALARY
	BERNHARDT	36000	500	36500
	COLLINS	28000	1000	29000
	MEIER	27000	800	27800
	SCHINDLER	18500	300	18800
	MARTINEZ	20500	500	21000
	PERRY	26000	1200	27200
	MERTA	21000	300	21300
	POSITIONS OF ACCOUNTANT	7	181600	
	LINCOLN	17000	500	17500
	HAMILTON	18000	600	18600
	GRIER	16750	550	17300
	THOMPSON	17000	550	17550
	STEVENS	21000	900	21900
	POSITIONS OF ACTUARY	5	92850	
	HALL	25000	20000	45000

## BEFORE BREAK PROCESSING

```

0010 *
0020 * This program retrieves all employees from a specific
0030 * department and summarizes the total employees in
0040 * each selected department.
0050 *
0060 DEFINE DATA LOCAL USING DDHLDA
0070 END-DEFINE
0080 READ EMPLOYEES BY DEPT = 'SYS' THRU 'SYSZ99'
0090   ACCEPT IF BIRTH GT 561231
0100   COMPRESS NAME ', ' INTO NAME LEAVING NO
0110   COMPRESS NAME FIRST-NAME #INITIAL INTO NAME
0120   DISPLAY NAME BIRTH 'DEPT' DEPT
0130   AT BREAK OF DEPT
0140     WRITE T*DEPT COUNT(DEPT) (SG=OFF EM=ZZZZZZ) /* (AD=L)
0150   END-BREAK
0160 END-READ
0170 END

```

NAME	DATE OF BIRTH	DEPT
LEISNER, SONJA T	59/02/14	SYSA01
RASMUSSEN, SUSANNE B	58/04/05	SYSA01
SOERENSEN, KARL	60/04/21	SYSA01
		3
		3
JENSEN, CLAUD J	64/05/07	SYSA03
ERIKSSEN, JONAS K	66/12/31	SYSA03
STEENSEN, RONNY S	58/11/24	SYSA03
STEGMANN, THIMES	60/04/03	SYSA03
BERTELSEN, ELISABETH	68/05/13	SYSA03
GUSTAVSEN, POUL O	59/03/04	SYSA03
		6
		6
		6
		6

- The ACCEPT causes records to be evaluated during the AT BREAK process when they are actually incorrect records. Notice the numerous “breaks” in the example output above. A better approach is illustrated on the next page.

**BEFORE BREAK PROCESSING**

```

0010 *
0020 * This program retrieves all employees from a specific
0030 * department and summarizes the total employees in
0040 * each selected department.
0050 *
0060 DEFINE DATA LOCAL USING DDHLDA
0070 END-DEFINE
0080 READ EMPLOYEES BY DEPT = 'SYS' THRU 'SYSZ99'
0090 BEFORE BREAK PROCESSING
0100     ACCEPT IF BIRTH GT 561231
0110 END-BEFORE
0120 COMPRESS NAME ', ' INTO NAME LEAVING NO
0130 COMPRESS NAME FIRST-NAME #INITIAL INTO NAME
0140 DISPLAY NAME BIRTH 'DEPT' DEPT
0150 AT BREAK OF DEPT
0160     WRITE T*DEPT COUNT(DEPT) (SG=OFF EM=ZZZZZ) /* (AD=L)
0170 END-BREAK
0180 END-READ
0190 END
    
```

MORE _			
Page	1		2005-02-01 12:34:56
	NAME	DATE OF BIRTH	DEPT
	-----	-----	-----
	LEISNER, SONJA T	59/02/14	SYSA01
	RASMUSSEN, SUSANNE B	58/04/05	SYSA01
	SOERENSEN, KARL	60/04/21	SYSA01
			3
	JENSEN, CLAUD J	64/05/07	SYSA03
	ERIKSSEN, JONAS K	66/12/31	SYSA03
	STEENSEN, RONNY S	58/11/24	SYSA03
	STEGMANN, THIMES	60/04/03	SYSA03
	BERTELSEN, ELISABETH	68/05/13	SYSA03
	GUSTAVSEN, POUL O	59/03/04	SYSA03
			6

**BEFORE BREAK PROCESSING**

```

DEFINE DATA LOCAL USING VEH-HIST
LOCAL                                /* The requirement was to collect totals on
1 #KEY (A17)                          /* the combined key of policy and its effective
1 REDEFINE #KEY                       /* date. In practice a simple AT BREAK was
  2 #POLNUM (A9)                       /* not taken when the policy effective date
  2 #EFFDATE(A8)                       /* was the same for two different policies
END-DEFINE                            /* appearing one after the other.
READ VEHICLES-HISTORY BY STATE-POLNUM-EFFDATE-VEHNUM = 'NY'
  BEFORE BREAK PROCESSING
    MOVE VEHICLE-POLICY-NUMBER TO #POLNUM
    MOVE VEHICLE-POLICY-EFF-DATE TO #EFFDATE
  END-BEFORE
  AT BREAK OF #KEY
    WRITE '***' COUNT(#KEY)
  END-BREAK
  DISPLAY RECORD-STATUS VEHICLE-POLICY-NUMBER
    VEHICLE-POLICY-EFF-DATE VEHICLE-FILE-NO
END-READ
END
    
```

RECORD STATUS	POLICY NUMBER	POLICY EFF-DATE	VEHICLE NUMBER
E	AIAL77504	19991001	9
E	AIAL77504	19991001	10
E	AIAL77504	19991001	12
E	AIAL77504	19991001	14
E	AIAL77504	19991001	16
E	AIAL77504	19991001	18
E	AIAL77504	19991001	20
E	AIAL77504	19991001	103
***	8		
V	AIAL77504	20001001	11
S	AIAL77504	20001001	121
***	2		
S	AIAL78187	20000724	1
***	1		
V	AIAX58120	20000228	4
***	9		
V	AIK810005	19991101	7
***	1		
V	AIK810005	20001101	6
S	AIK810005	20001101	6
V	AIK810005	20001101	8
V	AIK810005	20001101	9
V	AIK810005	20001101	10
S	AIK810005	20001101	10
***	6		

Notes